Lundi 5 janvier 2004 à 18 h

Art et mathématiques

Bernard Vuilleumier

Il existe de nombreux points de contact entre certaines œuvres artistiques et les mathématiques. La symétrie est un exemple banal de concept utilisé en art et en mathématiques. Mais on peut trouver d'autres notions, de portée aussi générale que la symétrie, qui sont communes aux deux domaines, par exemple la périodicité - ou plutôt la quasi-périodicité. Les seules formes d'art dans lesquelles ces notions apparaissaient n'ont longtemps été que la poésie et la musique. Les poètes et les compositeurs ont produit des œuvres traversées par la répétition d'un ou de plusieurs « motifs ». Mais aucune de ces répétitions n'est vraiment périodique, car un auteur répète raremement un passage exactement de la même manière. Ces compositions peuvent alors être qualifiées de quasi-périodiques.



Fig. 1 : Fonctions : périodique (à gauche) et quasi périodique (à droite). Le temps séparant deux maxima successifs est constant pour la fonction périodique alors qu'il ne l'est pas pour celle qui est quasi-périodique.

La notion de quasi-périodicité est bien connue en mathématiques. Elle apparaît par exemple dans des fonctions comme les fonctions de Bessel qui sont « presque » périodiques, alors que les fonctions trigonométriques sont rigoureusement périodiques. Quelles sont les différences entre ces deux classes d'objets mathématiques ? On peut définir une fonction en donnant l'équation différentielle qu'elle satisfait. Les fonctions trigonométriques strictement périodiques satisfont des équations linéaires à coefficients constants. Les fonctions quasi-périodiques de Bessel sont aussi solutions d'équations linéaires, mais à coefficients non constants. Cette incursion dans le domaine formel permet d'illustrer une démarche très fréquemment utilisée en science pour traiter des phénomènes non linéaires : remplacer les équations qui les décrivent par des équations linéaires mais dont les coefficients dépendent de la variable indépendante.



Fig. 2 : Ensembles : non connexe (à gauche) et connexe (à droite). Un ensemble non connexe peut être considéré comme un archipel et un ensemble connexe comme une île : tous les points peuvent être atteints par voie terrestre.

La non-linéarité, longtemps confinée au cercle restreint des spécialistes, est apparue depuis quelques années au grand public, associée à des termes comme « chaos », « attracteur étrange », « fractale », etc. Elle a donné lieu à une nouvelle forme d'art qui doit sa popularité à la beauté étrange d'images qui exercent une véritable fascination sur l'imaginaire. Les scientifiques s'y intéressent aussi de plus en plus car ils savent aujourd'hui que les phénomènes naturels relèvent, dans leur grande majorité, de la non-linéarité !

Prochaine réunion : lundi 2 février 2004 à 18 h

Travaux pratiques

Exercice 1

a) Comparez la fonction $y = \cos(x)$ à la fonction de Bessel $y = J_0(x)$ en traçant le graphique de ces deux fonctions sur le même système d'axes pour x compris entre 0 et 5π.
b) Tracez les graphiques des fonctions de Bessel $J_n(x)$ pour n compris entre 0 et 3 sur le même intervalle x.

Exercice 2

a) Donnez l'équation différentielle qui admet comme solution $y = a\cos(x)$. Cette équation est-elle linéaire ? Les coefficients de cette équation sont-ils constants ? Quels phénomènes cette équation permet-elle de décrire ?
b) Vérifiez que les fonctions $y = J_n(x)$ sont solutions de l'équation différentielle $x^2 y'' + x y' + (x^2 - n^2)\, y = 0$. Cette équation est-elle linéaire ? Les coefficients de cette équation sont-ils constants ? Dans quels domaines rencontre-t-on les fonctions qui sont solutions de cette équation ?

Exercice 3

Les fonctions quadratiques d'un nombre z et à un paramètre c, où z et c sont des nombres complexes, permettent d'obtenir des images d'une grande beauté et d'une étonnante variété lorsqu'elles sont itérées.
a) Ecrivez un programme comportant une boucle et permettant d'itérer une fonction quadratique $Q_c(z) = z^2 + c$ tant que le module de $z = x + iy$ est inférieur à 2 et que le nombre n d'itérations ne dépasse pas une limite n_{limite}.
b) Définissez une fonction $Q_c(x, y, n_{\text{limite}}, c_x, c_y)$ qui reçoit comme arguments la partie réelle x et la partie imaginaire y de z, le nombre maximum n_{limite} d'itérations, ainsi que la partie réelle c_x et imaginaire c_y de c, et qui retourne le nombre d'itérations effectuées.
c) Utilisez une instruction de représentation graphique à laquelle vous fournirez comme argument la fonction $Q_c(x, y, n_{\text{limite}}, c_x, c_y)$ et qui retournera, pour chaque valeur de x et y comprises dans un certain intervalle :
• un niveau de gris;
• une couleur;
• une altitude.
d) Dessinez le résultat pour les valeurs suivantes des arguments : $n_{\text{limite}} = 300$, $c_x = 0.360284$, $c_y = 0.100376$.
L'instruction de représentation graphique fera varier x et y dans les intervalles suivants : $-1 \le x \le 1$, $-1.2 \le y \le 1.2$.



■ Pour en savoir plus

• Martha L. Abell, James P. Braselton. *Differential Equations with Mathematica*. Academic Press, 1993.
• Michele Emmer, *The Visual Mind. Art and Mathematics*. The MIT Press. 1993.
• Robert L. Devaney. *Chaotic Dynamical Systems*. Second Edition. Addison-Wesley, 1989.

Art et mathématiques

Exercice 1 a)

La fonction $y = \cos(x)$ est rigoureusement périodique. La fonction de Bessel $y = J_0(x)$ est quasi-périodique.

```
Plot[{Cos[x], BesselJ[0, x]}, {x, 0, 5*Pi},
 GridLines → Automatic, Ticks → None, PlotStyle → {Hue[0.7], Hue[1]}]
```



Exercice 1 b)

```
Plot[Evaluate[Table[BesselJ[i, x], {i, 0, 3}]],
 {x, 0, 5*Pi}, GridLines → Automatic, Ticks → None]
```



Exercice 2

a) La fonction $y = a\cos(x)$ est solution de l'équation différentielle $\frac{d^2 y}{dx^2} + y = 0$. On le vérifie aisément de la manière suivante :

```
y[x_] := a*Cos[x]
D[y[x], {x, 2}] + y[x] == 0
```

```
True
```

La forme générale d'une équation différentielle ordinaire linéaire d'ordre n est la suivante :

$$a_n(x)\frac{d^n y}{dx^n} + a_{n-1}(x)\frac{d^{n-1} y}{dx^{n-1}} + \ldots + a_2(x)\frac{d^2 y}{dx^2} + a_1(x)\frac{dy}{dx} + a_0(x)y = f(x)$$

où $a_j(x)$, $j = 0, 1, \ldots, n$ et $f(x)$ sont donnés.

L' équation $\frac{d^2 y}{dx^2} + y = 0$ est donc une équation différentielle ordinaire linéaire d'ordre 2. Les coefficients $a_2 = 1$ et $a_0 = 1$ de cette équation ne dépendent pas de la variable indépendante x et sont constants. Cette équation modélise les oscillations harmoniques (oscillations provoquées par une force de rappel proportionnelle à l'écart par raport à la position d'équilibre).

b) Les fonctions de Bessel $y = J_n(x)$ sont solutions de l'équation $x^2\, y'' + x y' + (x^2 - n^2)\, y = 0$, comme on peut le vérifier, pour n'importe quelle valeur de n :

```
y[n_, x_] := BesselJ[n, x]
n = 0;
x^2*D[y[n, x], {x, 2}] + x*D[y[n, x], {x, 1}] + (x^2 - n^2) y[n, x] == 0 //
 FullSimplify
```

```
True
```

L'équation $x^2\, y'' + x y' + (x^2 - n^2)\, y = 0$ est linéaire mais ses coefficients $a_0 = (x^2 - n^2)$, $a_1 = x$ et $a_2 = x^2$ ne sont pas constants. On rencontre les fonctions de Bessel $y = J_n(x)$ aussi bien dans des problèmes de conduction de la chaleur que dans des problèmes de diffraction, d'acoustique, ou d'électromagnétisme.

Exercice 3

a) Tant que le module de $z = x + iy$ est inférieur à 2 et que le nombre n d'itérations ne dépasse pas n_{limite}, la boucle ci-dessous itère la fonction $Q_c(z) = z^2 + c$. Lorsque l'une ou l'autre des conditions n'est plus satisfaite, la boucle se termine et retounre le nombre n d'itérations effectuées :

```
While[Abs[z] < 2.0 && n <= nlim,
        z = z^2 + (cx + I*cy);
            ++n];
        n
```

b) Définissons la fonction $Q_c(x, y, n_{\text{limite}}, c_x, c_y)$

```
Qc = Compile[{x, y, nlim, cx, cy}, Module[{z, n = 0}, z = x + I * y;
    While[Abs[z] < 2.0 && n ≤ nlim, z = z^2 + (cx + I * cy);
         ++n];
     n]];
```

c) Les instructions de représentation graphique d'une fonction comportant deux variables x et y sont :

- **`Plot3D`**
- **`ContourPlot`**
- **`DensitiyPlot`**

Nous utiliserons **`DensitiyPlot`** qui permet d'obtenir des représentations en niveaux de gris ou en couleurs.

d) Utilisons **`DensityPlot`** en lui donnant comme argument la fonction $Q_c(x, y, n_{\text{limite}}, c_x, c_y)$ et en spécifiant le domaine de variation pour x et pour y :

```
DensityPlot[-Qc[x, y, 300, 0.360284, 0.100376], {x, -1, 1},
  {y, -1.2, 1.2}, PlotPoints → 300, Mesh → False, AspectRatio → Automatic,
  Axes → None, Frame → False, ColorFunction → (Hue[0.02 - #] &)];
```